Constructing Random Matrices with specified rank

Worksheet by Michael K. May, S.J., revised by Russell Blyth

```
> restart:  with(LinearAlgebra): with(plots): with(plottools):
Warning, the name changecoords has been redefined
Warning, the assigned name arrow now has a global binding
```

Outline

The basic objectives are:

1) Learn how to produce a random matrix of specified size and rank.
2) Explore how Maple finds a basis for the row space and column space of a matrix.
3) Relate the size of a matrix to the dimensions of the null space, the column space, and the row space.

```
>
```

Part 1: Random Matrices and Rank

As we test out theories in linear algebra it is often useful to produce a random matrix with specified size and rank. While testing theories out on such matrices is not a substitute for doing proofs, it can help in convincing us that it is worth looking for a proof. That is actually much of the battle in mathematics.

Recall that the Maple command `RandomMatrix(m,n);` creates an m by n matrix with entries randomly chosen integers between -100 and 100. (To choose your own range of permissible entries, specify the range using a third argument, for example: `RandomMatrix(m,n, generator=rand(-9..9));`). Since the rank of a matrix is the number of nonzero rows in the reduced echelon form of the matrix, it is bounded by the minimum of m and n.

```
> A := RandomMatrix(4,5);  Rank(A);
  B := RandomMatrix(6,3);  Rank(B);
```

$$A := \begin{bmatrix} -98 & -93 & -32 & 8 & 44 \\ -77 & -76 & -74 & 69 & 92 \\ 57 & -72 & -4 & 99 & -31 \\ 27 & -2 & 27 & 29 & 67 \end{bmatrix}$$

$$4$$

$$B := \begin{bmatrix} -80 & 50 & 45 \\ 43 & 10 & -81 \\ 25 & -16 & -38 \\ 94 & -9 & -18 \\ 12 & -50 & 87 \\ -2 & -22 & 33 \end{bmatrix}$$

3 (2.1)

It is worthwhile to note that the rank of a random matrix is usually actually equal to the minimum of m and n. To verify this we create 30 random matrices of random size, and for each matrix note the number of rows, the number of columns, and the rank; the third number in each triple (the rank of the matrix) in each case will almost always be the minimum of the first two numbers (number of rows and columns in the matrix) in the triple. [The Maple command rand(a..b)(); produces a random integer in the range [a,b].]

```
> RankList := [seq(0, j=1..30)]:
  for i from 1 to 30 do
     RowSize := rand(3..8)():
     ColSize := rand(3..8)():
     RankList[i] := [RowSize, ColSize, Rank(RandomMatrix(RowSize,
  ColSize))]:
  end do:
  RankList;
```

[[6, 3, 3], [5, 4, 4], [6, 7, 6], [4, 4, 4], [5, 8, 5], [4, 6, 4], [7, 5, 5], [7, 6, 6], [5, 8, 5], [6, 6, 6], [5, 3, 3], [5, 6, 5], [6, 4, 4], [7, 3, 3], [8, 7, 7], [4, 4, 4], [6, 8, 6], [6, 5, 5], [7, 8, 7], [5, 7, 5], [3, 6, 3], [4, 3, 3], [8, 5, 5], [6, 5, 5], [5, 4, 4], [5, 5, 5], [7, 8, 7], [5, 8, 5], [3, 3, 3], [6, 6, 6]] (2.2)

Adjoining rows or columns that are linear combinations of the existing rows and columns of a matrix does not change the rank. In keeping with the random way we are constructing matrices, we will adjoin random linear combinations of rows and columns to matrices to test this.

```
> bigA := <A,
     rand(-10..10)()*Row(A,1)+rand(-10..10)()*Row(A,2)+
        rand(-10..10)()*Row(A,3)+rand(-10..10)()*Row(A,4)>;
  [Rank(A),Rank(bigA)];
```

$$bigA := \begin{bmatrix} -98 & -93 & -32 & 8 & 44 \\ -77 & -76 & -74 & 69 & 92 \\ 57 & -72 & -4 & 99 & -31 \\ 27 & -2 & 27 & 29 & 67 \\ 778 & 763 & 187 & -32 & -371 \end{bmatrix}$$

[4, 4] (2.3)

```
> bigB := <B |
     rand(-10..10)()*Column(B,1) + rand(-10..10)()*Column(B,2) +

         rand(-10..10)()*Column(B,3)|
      rand(-10..10)()*Column(B,2) + rand(-10..10)()*Column(B,3)
```

```
>;
[Rank(B),Rank(bigB)];
```

$$bigB := \begin{bmatrix} -80 & 50 & 45 & -350 & -400 \\ 43 & 10 & -81 & -79 & -80 \\ 25 & -16 & -38 & -35 & 128 \\ 94 & -9 & -18 & 720 & 72 \\ 12 & -50 & 87 & 530 & 400 \\ -2 & -22 & 33 & 136 & 176 \end{bmatrix}$$

$$[3, 3] \tag{2.4}$$

```
>
```

Exercises:

1) Generate 3 random matrices of the following sizes: 3 x 4; 6 x 3; and 4 x 7. Test the rank of these matrices and verify that it is the minimum of the number of rows and the number of columns.

```
>
```

2) Explain why we generally expect a random m by n matrix to have rank either m or n, and not to have smaller rank (think, for example, about the subspace of R^3 spanned by three randomly chosen vectors in R^3).

```
>
```

3) Use Maple to produce a 4 by 6 matrix of rank 3 named MyMatrix.

```
>
>
>
```

Finding Bases for the Row and Column Spaces

The Maple commands RowSpace and ColumnSpace produce bases for the row space and the column space respectively. It should be noted that augmenting by linear combinations of columns does not change the column space and stacking linear combinations of rows does not change the row space.

```
> rbasA := RowSpace(A);
  rbasbigA := RowSpace(bigA);
  cbasB := ColumnSpace(B);
  cbasbigB := ColumnSpace(bigB);
```

$$rbasA := \left[\begin{bmatrix} 1 & 0 & 0 & 0 & \frac{-64334045}{22317012} \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 & \frac{22131413}{11158506} \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 & \frac{4483558}{1859751} \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 1 & \frac{64417651}{22317012} \end{bmatrix} \right]$$

$$rbasbigA := \left[\begin{bmatrix} 1 & 0 & 0 & 0 & \frac{-64334045}{22317012} \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 & \frac{22131413}{11158506} \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 & \frac{4483558}{1859751} \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 1 & \frac{64417651}{22317012} \end{bmatrix} \right]$$

$$cbasB := \left[\begin{bmatrix} 1 \\ 0 \\ 0 \\ -\frac{137141}{72320} \\ -\frac{10271}{9040} \\ -\frac{475}{1808} \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ \frac{18845}{14464} \\ -\frac{2105}{1808} \\ -\frac{1137}{1808} \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ -\frac{65799}{14464} \\ -\frac{2085}{1808} \\ \frac{291}{1808} \end{bmatrix} \right]$$

$$cbasbigB := \left[\begin{bmatrix} 1 \\ 0 \\ 0 \\ -\frac{137141}{72320} \\ -\frac{10271}{9040} \\ -\frac{475}{1808} \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ \frac{18845}{14464} \\ -\frac{2105}{1808} \\ -\frac{1137}{1808} \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ -\frac{65799}{14464} \\ -\frac{2085}{1808} \\ \frac{291}{1808} \end{bmatrix} \right] \qquad (3.1)$$

For the row space it is instructive to look at the reduced echelon form of the matrix, to see that its rows are used as a basis for the row space of the original matrix.

```
> ReducedRowEchelonForm(bigA);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & -\frac{64334045}{22317012} \\ 0 & 1 & 0 & 0 & \frac{22131413}{11158506} \\ 0 & 0 & 1 & 0 & \frac{4483558}{1859751} \\ 0 & 0 & 0 & 1 & \frac{64417651}{22317012} \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \qquad (3.2)$$

To find a basis for the column space it stands to reason that we want to choose the columns of a matrix that has been column reduced. To get Maple to do that for us we row reduce the transpose of the matrix and then transpose back.

```
> Transpose(ReducedRowEchelonForm(Transpose(bigB)));
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ -\frac{137141}{72320} & \frac{18845}{14464} & -\frac{65799}{14464} & 0 & 0 \\ -\frac{10271}{9040} & -\frac{2105}{1808} & -\frac{2085}{1808} & 0 & 0 \\ -\frac{475}{1808} & -\frac{1137}{1808} & \frac{291}{1808} & 0 & 0 \end{bmatrix} \tag{3.3}$$

Read off a basis for the column space from the first three columns of the result.

>

Exercises:

4) Find bases for the row space and column space of MyMatrix defined above.

>

5) Use Maple to row reduce and column reduce MyMatrix, and verify that the bases found in Exercise 4 are the nonzero rows or columns of the reduced matrices.

>

>

Connecting the Row Space and the Column Space

In the last two exercises you should have noticed that the dimensions of the row space and the column space of a matrix are equal. In both cases this corresponds to the number of pivots in reduced echelon form of the matrix, that is, the rank of the matrix. We have noted in class that the dimension of the null space is the number of free variables in the reduced matrix. In Chapter 4 we will discover an important relationship between the vectors in the row space and the vectors in the null space (recall that both spaces are subspaces of R^n if A is an m x n matrix). This relationship involves the dot product extended naturally to R^n. We check dot products to note that the vectors in the null space are orthogonal to the vectors in the row space (two vectors are orthogonal if their dot product is zero).

```
> rbasbigA := RowSpace(bigA);
  nbasbigA := NullSpace(bigA);
```

$$rbasbigA := \left[\left[1 \quad 0 \quad 0 \quad 0 \quad -\frac{64334045}{22317012} \right], \left[0 \quad 1 \quad 0 \quad 0 \quad \frac{22131413}{11158506} \right], \left[0 \quad 0 \quad 1 \quad 0 \quad \frac{4483558}{1859751} \right], \left[0 \quad 0 \quad 0 \quad 1 \quad \frac{64417651}{22317012} \right] \right]$$

$$nbasbigA := \left\{ \begin{bmatrix} \frac{64334045}{22317012} \\ -\frac{22131413}{11158506} \\ -\frac{4483558}{1859751} \\ -\frac{64417651}{22317012} \\ 1 \end{bmatrix} \right\} \quad (4.1)$$

```
> rbasbigA[1].nbasbigA[1];
  rbasbigA[2].nbasbigA[1];
  rbasbigA[3].nbasbigA[1];
  rbasbigA[4].nbasbigA[1];
```

$$0$$
$$0$$
$$0$$
$$0 \quad (4.2)$$

\>

Exercises:

6) Verify that the vectors in a basis for the row space of MyMatrix are orthogonal to vectors in a basis for the nullspace.

\>

7) If we stack a random m by n matrix on top of a basis for its null space and use Gauss-Jordan elimination, what can you say about the the number of pivots and free variables in the resulting matrix?

\>

\>

\>